

drv.mp

derivation trees with METAPOST*

almost a user guide†

Laurent M´
laurent.mehats@gmail.com

documented version: 0.92

$$\frac{\frac{\frac{\overline{\overline{A, \Gamma \vdash B}}^\gamma}{A, \Gamma, \Delta \vdash B \wedge C} \wedge_R \quad \frac{\overline{\Delta \vdash C}^\delta}{B \wedge C, \Theta \vdash D}^\theta}{\frac{\frac{\frac{\overline{\overline{A, \Gamma, \Delta, \Theta \vdash D}}}{\Gamma, \Delta, \Theta \vdash A \rightarrow D} \rightarrow_R \quad \frac{\overline{\overline{E, \Upsilon \vdash F}}^\nu}{\Gamma, \Delta, \Theta, (A \rightarrow D) \rightarrow E, \Upsilon \vdash F} \rightarrow_L}{\frac{\overline{\Pi \vdash (A \rightarrow D) \rightarrow E}^\pi}{\Gamma, \Delta, \Theta, \Pi, \Upsilon \vdash F} \text{cut}} \text{cut}$$

Contents

1	Usage	2
1.1	Structure of a METAPOST file using <code>drv.mp</code>	2
1.2	Running METAPOST	3
1.3	L ^A T _E X inclusion commands	3
2	Judgment & inference declarations	4
2.1	<code>jgm</code> & <code>nfr</code>	4
2.2	<code>dcl</code>	6
2.3	<code>mvd</code>	7
2.4	Sub-tree delimiters & labels	9

*You don't need to know METAPOST to use this package.

†Feel free to improve! Last update: June 9, 2009.

3	drv tunings	10
3.1	drv_font_size	10
3.2	drv_math_style	10
3.3	drv_scale	11
3.4	drv_junction_style	12
3.5	drv_alignment_style	12
3.6	drv_labels_position	13
3.7	drv_roots_position	13
3.8	drv_axis_reference	13
3.9	drv_left_delimiter & drv_right_delimiter	13
3.10	drv_fraction_mode	14
3.11	drv_proof_mode	15
4	Low level features	15
4.1	Low level inference declarations	15
4.2	Inside derivation trees	17
4.3	Basic METAPOST use	19
	References	21
A	Debugging	21
B	Related packages	22

1 Usage

1.1 Structure of a METAPOST file using drv.mp

Preamble

```
input drv;
verbatimtex %&latex
<LATEX preamble>
\begin{document}
etex;
```

Figures

```
<optional drv tunings>
beginfig(<index>)
  <judgment & inference declarations>
  draw drv_tree;
  <optional extra METAPOST code>
endfig;
```


Postamble

end

For each “`\beginfig(<index>), \endfig;`” pair in a file `<jobname>.mp`, METAPOST will output an Encapsulated PostScript file `<jobname>.<index>`.

1.2 Running METAPOST

You have to run

```
mpost <jobname>.mp
```

at least twice (once more if you use sub-tree delimiters, see § 2.4). On the first run, METAPOST collects the L^AT_EX code generated by `drv.mp` declaration macros (see § 2) and writes it to the file `<jobname>-delayed.mp`. On the second run, METAPOST preprocesses the L^AT_EX code in `<jobname>-delayed.mp` and typesets the derivation trees.

If you get an error on the first run then it comes from the `drv.mp`/METAPOST code. If you get an error on the second run then it comes from the L^AT_EX code. In both cases, correct the error (see § A), delete `<jobname>-delayed.mp` and run “`mpost <jobname>.mp`” twice again (a `makefile` can do that for you).

1.3 L^AT_EX inclusion commands

Encapsulated PostScript files `<jobname>.<index>` generated by METAPOST can be included in L^AT_EX documents using the `\includegraphics{<jobname>.<index>}` command from the `graphics.sty` (or `graphicx.sty`) package.

However, I suggest using the following `\drv{<jobname>.<index>}` command, which is such that the baseline of the included picture coincides with the baseline of the inclusion point.

```
\usepackage{graphicx}
\makeatletter
\def\Gin@def@bp#1\relax#2#3{\gdef#2{#3}}
\newsavebox{\graphicsbox}
\newcommand*{\drv}[1]{%
\sbox{\graphicsbox}{\includegraphics{#1}}%
\raisebox{\Gin@lly bp}{%
{\usebox{\graphicsbox}}}}
\makeatother
```

The code for `\drv` was suggested by Josselin N on the `fr.comp.text.tex` Usenet group.

2 Judgment & inference declarations

2.1 jgm & nfr

`jgm` $\langle \text{numeric} \rangle$ $\langle \text{strings list} \rangle$
 $\langle \text{numeric} \rangle$ judgment index
 $\langle \text{strings list} \rangle$ sub-judgments *math-mode* L^AT_EX code
`nfr` $\langle \text{numeric 1} \rangle$ $(\langle \text{numerics list} \rangle)$ $(\langle \text{string} \rangle, \langle \text{numeric 2} \rangle)$
 $\langle \text{numeric 1} \rangle$ inference index
 $\langle \text{numerics list} \rangle$ list of premise indices
 $\langle \text{string} \rangle$ inference label *math-mode* L^AT_EX code
 $\langle \text{numeric 2} \rangle$ inference line style (0, 1, 2, 3, 4, 5 or 6)

“`jgm` $\langle \text{numeric} \rangle$ ” declares a judgment which index is $\langle \text{numeric} \rangle$; “`nfr` $\langle \text{numeric} \rangle$ ” declares an inference which conclusion is the index $\langle \text{numeric} \rangle$ judgment. A judgment can be declared before or after the corresponding inference, no matter.

First example

```

beginfig(110)
jgm 0 "A\vdash B";
jgm 1 "B\vdash C";
jgm 2 "A\vdash C";
nfr 0 () ("f", 1);
nfr 1 () ("g", 1);
nfr 2 (0, 1) ("\circ", 1);
draw drv_tree;
endfig;

```

$$\frac{\overline{A \vdash B}^f \quad \overline{B \vdash C}^g}{A \vdash C}^o$$

Inference line styles

```

beginfig(111)
jgm 10 "\text{none}";
jgm 11 "\text{simple}";
jgm 12 "\text{double}";
jgm 13 "\text{dotted}";
jgm 14 "\text{dashed}";
jgm 15 "\text{waved}";
jgm 16 "\text{\TeX-dotted}";
nfr 10 () ("\leftarrow 0", 0);
nfr 11 (10) ("\leftarrow 1", 1);
nfr 12 (11) ("\leftarrow 2", 2);
nfr 13 (12) ("\leftarrow 3", 3);
nfr 14 (13) ("\leftarrow 4", 4);
nfr 15 (14) ("\leftarrow 5", 5);
nfr 16 (15) ("\leftarrow 6", 6);
draw drv_tree;
endfig;

```

$$\begin{array}{l}
\leftarrow 0 \\
\text{none} \\
\hline
\text{simple} \quad \leftarrow 1 \\
\hline
\text{double} \quad \leftarrow 2 \\
\hline
\text{dotted} \quad \leftarrow 3 \\
\hline
\text{dashed} \quad \leftarrow 4 \\
\hline
\text{waved} \quad \leftarrow 5 \\
\hline
\text{\TeX-dotted} \quad \leftarrow 6
\end{array}$$

Declarations order Declarations can occur in any order.

<pre> beginfig(120) % preorder declarations jgm 0 "0"; jgm 1 "00"; jgm 2 "000"; jgm 3 "001"; jgm 4 "002"; jgm 5 "01"; jgm 6 "010"; jgm 7 "011"; jgm 8 "012"; jgm 9 "02"; jgm 10 "020"; jgm 11 "021"; jgm 12 "022"; nfr 0 (1, 5, 9) ("a", 1); nfr 1 (2, 3, 4) ("b", 1); nfr 2 () ("c", 1); nfr 3 () ("d", 1); nfr 4 () ("e", 1); nfr 5 (6, 7, 8) ("f", 1); nfr 6 () ("g", 1); nfr 7 () ("h", 1); nfr 8 () ("i", 1); nfr 9 (10, 11, 12) ("j", 1); nfr 10 () ("k", 1); nfr 11 () ("l", 1); nfr 12 () ("m", 1); draw drv_tree; endfig; </pre>	<pre> beginfig(121) % postorder declarations jgm 0 "000"; jgm 1 "001"; jgm 2 "002"; jgm 3 "00"; jgm 4 "010"; jgm 5 "011"; jgm 6 "012"; jgm 7 "01"; jgm 8 "020"; jgm 9 "021"; jgm 10 "022"; jgm 11 "02"; jgm 12 "0"; nfr 0 () ("a", 1); nfr 1 () ("b", 1); nfr 2 () ("c", 1); nfr 3 (0, 1, 2) ("d", 1); nfr 4 () ("e", 1); nfr 5 () ("f", 1); nfr 6 () ("g", 1); nfr 7 (4, 5, 6) ("h", 1); nfr 8 () ("i", 1); nfr 9 () ("j", 1); nfr 10 () ("k", 1); nfr 11 (8, 9, 10) ("l", 1); nfr 12 (3, 7, 11) ("m", 1); draw drv_tree; endfig; </pre>
---	---

$$\begin{array}{ccccccc}
 \overline{000}^c & \overline{001}^d & \overline{002}^e & \overline{010}^g & \overline{011}^h & \overline{012}^i & \overline{020}^k & \overline{021}^l & \overline{022}^m \\
 \hline
 & 00 & & 01 & & 02 & & & \\
 \hline
 & & & 0 & & & & &
 \end{array} \quad (120)$$

$$\begin{array}{ccccccc}
 \overline{000}^a & \overline{001}^b & \overline{002}^c & \overline{010}^e & \overline{011}^f & \overline{012}^g & \overline{020}^i & \overline{021}^j & \overline{022}^k \\
 \hline
 & 00 & & 01 & & 02 & & & \\
 \hline
 & & & 0 & & & & &
 \end{array} \quad (121)$$

Sub-judgments

```

beginfig(130)
jgm 0 "a\otimes b";
resp. jgm 0 "a", "\otimes", "b";
nfr 0 () ("c", 1);
draw drv_tree;
endfig;

```

$$\frac{}{a \otimes b}^c$$

$$\frac{}{a \otimes b}^c$$

The outputs induced by

“jgm 0 "a\otimes b";” and “jgm 0 "a", "\otimes", "b";”

are the same. Using the latter declaration, you can manipulate sub-judgments independently from each-other (see § 4.2 and § 4.3).

Title page derivation tree

```

beginfig(100)
jgm 0 "\Pi\vdash (A\to D)\to E";
jgm 1 "A, \Gamma\vdash B";
jgm 2 "\Delta\vdash C";
jgm 3 "A, \Gamma, \Delta\vdash B\wedge C";
jgm 4 "B\wedge C, \Theta\vdash D";
jgm 5 "A, \Gamma, \Delta, \Theta\vdash D";
jgm 6 "\Gamma, \Delta, \Theta\vdash A\to D";
jgm 7 "E, \Upsilon\vdash F";
jgm 8 "\Gamma, \Delta, \Theta, (A\to D)\to E, \Upsilon\vdash F";
jgm 9 "\Gamma, \Delta, \Theta, \Pi, \Upsilon\vdash F";
nfr 0 () ("\pi", 4);
nfr 1 () ("\gamma", 2);
nfr 2 () ("\delta", 1);
nfr 3 (1, 2) ("\wedge_R", 1);
nfr 4 () ("\theta", 3);
nfr 5 (3, 4) ("\text{cut}", 1);
nfr 6 (5) ("\to_R", 1);
nfr 7 () ("\upsilon", 2);
nfr 8 (6, 7) ("\to_L", 1);
nfr 9 (0, 8) ("\text{cut}", 1);
draw drv_tree;
endfig;

```

2.2 dcl

dcl enables the simultaneous declarations of a judgment and of the corresponding inference.

“`dcl <numeric 1> (<numerics list>) (<string>, <numeric 2>) <strings list>;`”

is a shorthand for:

“`jgm <numeric 1> <strings list>;`
`nfr <numeric 1> (<numerics list>) (<string>, <numeric 2>);`”

```
beginfig(140)
dcl 0 () ("f", 1) "A\vdash B";
dcl 1 () ("g", 1) "B\vdash C";
dcl 2 (0, 1) ("\circ", 1) "A\vdash C";
draw drv_tree;
endfig;
```

$$\frac{\overline{A \vdash B}^f \quad \overline{B \vdash C}^g}{A \vdash C}^{\circ}$$

```
beginfig(141)
dcl 0 (1, 5, 9) ("a", 1) "0";
dcl 1 (2, 3, 4) ("b", 1) "00";
dcl 2 () ("c", 1) "000";
dcl 3 () ("d", 1) "001";
dcl 4 () ("e", 1) "002";
dcl 5 (6, 7, 8) ("f", 1) "01";
dcl 6 () ("g", 1) "010";
dcl 7 () ("h", 1) "011";
dcl 8 () ("i", 1) "012";
dcl 9 (10, 11, 12) ("j", 1) "02";
dcl 10 () ("k", 1) "020";
dcl 11 () ("l", 1) "021";
dcl 12 () ("m", 1) "022";
draw drv_tree;
endfig;
```

$$\frac{\frac{\overline{000}^c \quad \overline{001}^d \quad \overline{002}^e}{00}^b \quad \frac{\overline{010}^g \quad \overline{011}^h \quad \overline{012}^i}{01}^f \quad \frac{\overline{020}^k \quad \overline{021}^l \quad \overline{022}^m}{02}^j}{0}^a$$

2.3 mvd


A premise index *<numeric 1>* in an inference declaration can be replaced with

“`mvd <numeric 1> (<numeric 2>, <numeric 3>)`”


which declares *<numeric 2>* “phantom” inference steps starting from judgment *<numeric 1>*. The “phantom” inference steps are intended to be drawn as a path using path-style *<numeric 3>*.

`mvd` $\langle \text{numeric } 1 \rangle$ ($\langle \text{numeric } 2 \rangle$, $\langle \text{numeric } 3 \rangle$)
 $\langle \text{numeric } 1 \rangle$ index of the origin judgment
 $\langle \text{numeric } 2 \rangle$ number of steps
 $\langle \text{numeric } 3 \rangle$ phantom steps style (0, 1, 2, 3, 4, 5 or 6)

```
beginfig(150)
jgm 1 "aaa";
jgm 2 "bbb";
jgm 3 "ccc";
jgm 4 "d";
nfr 1 () ("", 1);
nfr 2 () ("", 1);
nfr 3 () ("", 1);
nfr 4 (mvd 1 (2, 3), 2, 3) ("", 1);
resp. nfr 4 (1, mvd 2 (2, 4), 3) ("", 1);
resp. nfr 4 (1, 2, mvd 3 (2, 5)) ("", 1);
draw drv_tree;
endfig;
```



```
beginfig(160)
jgm 0 "\textsc{Size matters -- Part 1}";
jgm 1 "\text{Here is a rather long"& % string concatenation
      " judgment that I don't want to shorten.}";
jgm 2 "\text{Will the derivation tree fit on the page?}";
jgm 3 "\text{It does.}";
nfr 0 () ("", 0);
nfr 1 (0) ("", 1);
nfr 2 () ("", 1);
nfr 3 (mvd 1 (2, 3), 2) ("", 1);
draw drv_tree;
endfig;
```



2.4 Sub-tree delimiters & labels

- **Nfr** $\langle \text{numeric } I \rangle$ ($\langle \text{numerics list} \rangle$)
 $(\langle \text{string } 1 \rangle, \langle \text{string } 2 \rangle, \langle \text{string } 3 \rangle, \langle \text{numeric } 2 \rangle)$
 - $\langle \text{numeric } I \rangle$ inference index
 - $\langle \text{numerics list} \rangle$ list of premise indices
 - $\langle \text{string } 1 \rangle$ inference label *math-mode* \LaTeX code
 - $\langle \text{string } 2 \rangle$ left delimiter label *math-mode* \LaTeX code
 - $\langle \text{string } 3 \rangle$ right delimiter label *math-mode* \LaTeX code
 - $\langle \text{numeric } 2 \rangle$ inference line style (0, 1, 2, 3, 4, 5 or 6)

```

beginfig(170)
jgm 1 "a";
jgm 2 "b";
jgm 3 "c";
jgm 4 "d";
nfr 1 () ("", 1);
nfr 2 () ("", 1);
Nfr 3 (1, 2) ("", "E", "", 1);
Nfr 4 (3) ("", "", "F", 1);
draw drv_tree;
endfig;

```

$$E \left\{ \frac{\overline{a} \ \overline{b}}{\frac{c}{d}} \right\} F$$

- “**Dcl** $\langle \text{numeric } I \rangle$ ($\langle \text{numerics list} \rangle$)
 $(\langle \text{string } 1 \rangle, \langle \text{string } 2 \rangle, \langle \text{string } 3 \rangle, \langle \text{numeric } 2 \rangle) \langle \text{strings list} \rangle;$ ”
is a shorthand for:
“**jgm** $\langle \text{numeric } I \rangle$ $\langle \text{strings list} \rangle;$
Nfr $\langle \text{numeric } I \rangle$ ($\langle \text{numerics list} \rangle$)
 $(\langle \text{string } 1 \rangle, \langle \text{string } 2 \rangle, \langle \text{string } 3 \rangle, \langle \text{numeric } 2 \rangle);$ ”

```

beginfig(180)
dcl 1 () ("", 1) "a";
Dcl 2 (1) ("", "", "B", 1) "c";
dcl 3 () ("", 1) "d";
Dcl 4 (2, 3) ("", "E", "", 1) "f";
draw drv_tree;
endfig;

```

$$E \left\{ \frac{\overline{a}}{c} \right\}^B \frac{\overline{d}}{f}$$

- **Mvd** $\langle \text{numeric } 1 \rangle$ ($\langle \text{numeric } 2 \rangle$, $\langle \text{string } 1 \rangle$, $\langle \text{string } 2 \rangle$, $\langle \text{numeric } 3 \rangle$)
 - $\langle \text{numeric } 1 \rangle$ index of the origin judgment
 - $\langle \text{numeric } 2 \rangle$ number of inference steps
 - $\langle \text{string } 1 \rangle$ left phantom steps label *math-mode* \LaTeX code
 - $\langle \text{string } 2 \rangle$ right phantom steps label *math-mode* \LaTeX code
 - $\langle \text{numeric } 3 \rangle$ phantom steps style (0, 1, 2, 3, 4, 5 or 6)

```

beginfig(190)
jgm 1 "aaa";
jgm 2 "bbb";
jgm 3 "ccc";
nfr 1 () ("", 1);
nfr 2 () ("", 1);
nfr 3 (Mvd 1 (2, "d", "", 3), 2) ("", 1);
draw drv_tree;
endfig;

```

$$\begin{array}{c}
\overline{aaa} \\
\vdots \\
d \quad \overline{bbb} \\
\hline
ccc
\end{array}$$

3 drv tunings

3.1 drv_font_size

drv_font_size $\langle \text{string} \rangle$
 $\langle \text{string} \rangle$ \LaTeX font-size command
 "\tiny"
 "\scriptsize"
 "\footnotesize"
 "\small"
 "\normalsize" * default *
 "\large"
 etc.

$$\begin{array}{c}
\text{"\tiny"} \\
4 \left\{ \frac{\overline{a}^1 \overline{b}^2}{c} \right\}_3 \\
\text{"\small"} \\
4 \left\{ \frac{\overline{a}^1 \overline{b}^2}{c} \right\}_3 \\
\text{"\Large"} \\
4 \left\{ \frac{\overline{a}^1 \overline{b}^2}{c} \right\}_3
\end{array}$$

3.2 drv_math_style

drv_math_style ($\langle \text{suffix} \rangle$, $\langle \text{string} \rangle$)
 $\langle \text{suffix} \rangle$ component type (drv, jdg, ilb, dlb or plb)
 drv derivation tree * default style: "\displaystyle" *
 jdg judgment * default style: "\textstyle" *
 ilb inference label * default style: "\scriptstyle" *
 dlb delimiter label * default style: "\textstyle" *
 plb phantom steps label * default style: "\textstyle" *
 $\langle \text{string} \rangle$ \LaTeX math-style command

Examples “drv_math_style (drv, —);”

"\displaystyle"	"\textstyle"	"\scriptstyle"
$4 \left\{ \frac{\overline{a}^1 \overline{b}^2}{c} \right\}_3$	$4 \left\{ \frac{\overline{a}^1 \overline{b}^2}{c} \right\}_3$	$4 \left\{ \frac{\overline{a}^1 \overline{b}^2}{c} \right\}_3$

Examples “drv_math_style (jdg, —);”

"\displaystyle"	"\textstyle"	"\scriptstyle"
$4 \left\{ \frac{\overline{\bigwedge_{i \in I} A_i}^1 \overline{b}^2}{c} \right\}_3$	$4 \left\{ \frac{\overline{\bigwedge_{i \in I} A_i}^1 \overline{b}^2}{c} \right\}_3$	$4 \left\{ \frac{\overline{\bigwedge_{i \in I} A_i}^1 \overline{b}^2}{c} \right\}_3$

Examples “drv_math_style (ilb, —);”

"\textstyle"	"\scriptstyle"	"\scriptscriptstyle"
$\frac{\overline{a}^1 \overline{b}^2}{c} \bigg _3$	$\frac{\overline{a}^1 \overline{b}^2}{c} \bigg _3$	$\frac{\overline{a}^1 \overline{b}^2}{c} \bigg _3$

3.3 drv_scale

drv_scale (<suffix>, <numeric>)

<suffix> scale type (clr, prn, jdg or ilb)

clr	(see examples)	* default scale: 1 *
prn	(see examples)	* default scale: 1 *
jdj	(see examples)	* default scale: 1 *
ilb	(see examples)	* default scale: 1 *

<numeric> scale value

Examples “drv_scale (clr, —);”

0	1	2	4
$\frac{\overline{(a)}}{a}$	$\frac{\overline{(a)}}{a}$	$\frac{\overline{(a)}}{a}$	$\frac{\overline{(a)}}{a}$

Examples “drv_scale (prm, —);”

$$\begin{array}{cccc}
 \text{0} & \text{1} & \text{2} & \text{4} \\
 \hline
 \frac{a}{a} & \frac{a}{a} & \frac{a}{a} & \frac{a}{a} \\
 \hline
 a & a & a & a
 \end{array}$$

Examples “drv_scale (jgm, —);”

$$\begin{array}{cccc}
 \text{0} & \text{1} & \text{2} & \text{4} \\
 \hline
 \frac{a}{a} & \frac{a}{a} & \frac{a}{a} & \frac{a}{a} \\
 \hline
 a & a & a & a
 \end{array}$$

Examples “drv_scale (ilb, —);”

$$\begin{array}{cccc}
 \text{0} & \text{1} & \text{2} & \text{4} \\
 \hline
 \frac{-b}{a} & \frac{-b}{a} & \frac{-b}{a} & \frac{-b}{a} \\
 \hline
 a & a & a & a
 \end{array}$$

3.4 drv_junction_style

drv_junction_style $\langle \text{numeric} \rangle$
 $\langle \text{numeric} \rangle$ junction style (0, 1 or 2)
 0 “fully-interlacing”
 1 “semi-interlacing” * default *
 2 “non-interlacing”

$$\begin{array}{ccc}
 \text{0} & \text{1} & \text{2} \\
 \hline
 \frac{aaaaaaaaaa}{a} & \frac{aaaaaaaaaa}{a} & \frac{aaaaaaaaaa}{a} \\
 \hline
 \frac{a}{a} & \frac{a}{a} & \frac{a}{a} \\
 \hline
 \frac{a}{a} & \frac{aaaaaaa}{a} & \frac{aaaaaaa}{a} \\
 \hline
 \frac{a}{aaaaa} & \frac{a}{aaaaa} & \frac{a}{aaaaa} \\
 \hline
 a & a & a
 \end{array}$$

3.5 drv_alignment_style

drv_alignment_style $\langle \text{suffix} \rangle$
 $\langle \text{suffix} \rangle$ alignment style (l, c or r)
 l left
 c centered * default *
 r right

$$\begin{array}{c}
 \text{l} \\
 \hline
 \overline{a} \quad \overline{a} \quad \overline{a} \quad \overline{a} \\
 \hline
 \overline{a} \quad \overline{a} \\
 \hline
 a
 \end{array}
 \qquad
 \begin{array}{c}
 \text{c} \\
 \hline
 \overline{a} \quad \overline{a} \quad \overline{a} \quad \overline{a} \\
 \hline
 \overline{a} \quad \overline{a} \\
 \hline
 a
 \end{array}
 \qquad
 \begin{array}{c}
 \text{r} \\
 \hline
 \overline{a} \quad \overline{a} \quad \overline{a} \quad \overline{a} \\
 \hline
 \overline{a} \quad \overline{a} \\
 \hline
 a
 \end{array}$$

3.6 drv_labels_position

`drv_labels_position` $\langle suffix \rangle$
 $\langle suffix \rangle$ position (l or r)
 l left
 r right ** default **

$$\begin{array}{c}
 \overline{b} \quad \overline{b} \\
 \overline{a} \quad \overline{a} \\
 \hline
 a
 \end{array}
 \qquad
 \begin{array}{c}
 \overline{b} \quad \overline{b} \\
 \overline{a} \quad \overline{a} \\
 \hline
 a
 \end{array}$$

3.7 drv_roots_position

`drv_roots_position` $\langle suffix \rangle$
 $\langle suffix \rangle$ position (t or b)
 t top
 b bottom ** default **

$$\begin{array}{c}
 \overline{a} \\
 \hline
 \overline{a} \quad \overline{a} \\
 \hline
 \overline{a} \quad \overline{a} \quad \overline{a}
 \end{array}
 \qquad
 \begin{array}{c}
 \overline{a} \quad \overline{a} \quad \overline{a} \quad \overline{a} \\
 \hline
 \overline{a} \quad \overline{a} \\
 \hline
 a
 \end{array}$$

3.8 drv_axis_reference

`drv_axis_reference` $\langle suffix \rangle$
 $\langle suffix \rangle$ reference type (iln or jdg)
 iln root inference line ** default **
 jdg root judgment

$$\text{---math axis---} \frac{\overline{a} \quad \overline{a}}{a} \frac{\overline{a} \quad \overline{a}}{a} \text{---}$$

Notice that `drv_axis_reference` is irrelevant if you don't use the `\drv` inclusion command (see § 1.3).

3.9 drv_left_delimiter & drv_right_delimiter

`drv_left_delimiter` $\langle string \rangle$
 $\langle string \rangle$ left delimiter math-mode L^AT_EX code
 "(" (
 "[" [
 "\lbrace" { ** default **
 "\langle" <
 etc.

`drv_right_delimiter` $\langle string \rangle$
 $\langle string \rangle$ right delimiter math-mode L^AT_EX code
`"")`
`""]`
`"\rbrace"` } * *default* *
`"\rangle"` }
 etc.

Examples “`drv_left_delimiter —;`”

$$\begin{array}{ccc}
 \text{"("} & \text{"\lfloor floor"} & \text{"."} \\
 E \left\{ \frac{\overline{a}}{c} \right\} B \frac{\overline{d}}{f} & E \left[\frac{\overline{a}}{c} \right] B \frac{\overline{d}}{f} & E \frac{\overline{a}}{c} \left\{ B \frac{\overline{d}}{f} \right.
 \end{array}$$

Examples “`drv_right_delimiter —;`”

$$\begin{array}{ccc}
 \text{"\rangle"} & \text{"\uparrow"} & \text{"."} \\
 E \left\{ \frac{\overline{a}}{c} \right\} B \frac{\overline{d}}{f} & E \left\{ \frac{\overline{a}}{c} \uparrow B \frac{\overline{d}}{f} \right. & E \left\{ \frac{\overline{a}}{c} B \frac{\overline{d}}{f} \right.
 \end{array}$$

3.10 drv_fraction_mode

`drv_fraction_mode` $\langle suffix \rangle$
 $\langle suffix \rangle$ fraction-mode status (on or off)
 on * *default* *
 off

When derivation trees are typeset in “fraction mode”, heights of leaf-judgments above which there is no inference line are ignored. Warning: this mode may cause overlaps in conjunction with interlacing junction-styles (0 and 1).

on	off	typeset as
$ \frac{\overbrace{A, \Gamma \vdash B} \quad \overline{B, \Delta \vdash C}}{A, \Gamma, \Delta \vdash C} \text{cut} \quad \frac{\quad}{\Gamma, \Delta \vdash A \rightarrow C} \rightarrow_R $	$ \frac{\overbrace{A, \Gamma \vdash B} \quad \overline{B, \Delta \vdash C}}{A, \Gamma, \Delta \vdash C} \text{cut} \quad \frac{\quad}{\Gamma, \Delta \vdash A \rightarrow C} \rightarrow_R $	$ \frac{\overbrace{A, \Gamma \vdash B} \quad \overline{B, \Delta \vdash C}}{A, \Gamma, \Delta \vdash C} \text{cut} \quad \frac{\quad}{\Gamma, \Delta \vdash A \rightarrow C} \rightarrow_R $

3.11 drv_proof_mode

drv_proof_mode $\langle \text{suffix} \rangle$

$\langle \text{suffix} \rangle$ proof-mode status (on or off)

on

off * default *

$$\begin{array}{c}
 \text{on} \\
 \frac{\frac{\frac{\frac{\textcolor{red}{0}A}{\textcolor{blue}{0}} \quad \frac{\textcolor{red}{1}A}{\textcolor{blue}{1}}}{\textcolor{red}{2}A}{\textcolor{blue}{2}} \quad \frac{\textcolor{red}{3}B}{\textcolor{blue}{3}}}{\textcolor{red}{4}A}{\textcolor{blue}{4}} \quad \frac{\textcolor{red}{5}B}{\textcolor{blue}{5}}}{\textcolor{red}{6}A}{\textcolor{blue}{6}} \quad \frac{\textcolor{red}{1}B}{\textcolor{blue}{1}} \quad \frac{\textcolor{red}{2}B}{\textcolor{blue}{2}}}{\textcolor{red}{3}A}{\textcolor{blue}{3}} \quad \frac{\textcolor{red}{4}B}{\textcolor{blue}{4}}}{\textcolor{red}{5}A}{\textcolor{blue}{5}} \quad \frac{\textcolor{red}{6}B}{\textcolor{blue}{6}}} \textcolor{red}{-o}_L \\
 \frac{\textcolor{red}{3}A}{\textcolor{blue}{3}} \quad \frac{\textcolor{red}{4}B}{\textcolor{blue}{4}}}{\textcolor{red}{5}A}{\textcolor{blue}{5}} \quad \frac{\textcolor{red}{6}B}{\textcolor{blue}{6}}} \textcolor{red}{-o}_R
 \end{array}
 \qquad
 \begin{array}{c}
 \text{off} \\
 \frac{\frac{\textcolor{red}{1}A}{\textcolor{blue}{1}} \quad \frac{\textcolor{red}{2}B}{\textcolor{blue}{2}}}{\textcolor{red}{3}A}{\textcolor{blue}{3}} \quad \frac{\textcolor{red}{4}B}{\textcolor{blue}{4}}} \textcolor{red}{-o}_L \\
 \frac{\textcolor{red}{3}A}{\textcolor{blue}{3}} \quad \frac{\textcolor{red}{4}B}{\textcolor{blue}{4}}} \textcolor{red}{-o}_R
 \end{array}$$

Red numbers (resp. dots): judgment indices (resp. central points).

Blue numbers (resp. dots): sub-judgment indices (resp. central points).

4 Low level features

4.1 Low level inference declarations

- NFR $\langle \text{numeric } 1 \rangle$ ($\langle \text{numerics list} \rangle$)
 $(\langle \text{string } 1 \rangle, \langle \text{string } 2 \rangle, \langle \text{string } 3 \rangle, \langle \text{string } 4 \rangle)$
 $(\langle \text{numeric } 2 \rangle, \langle \text{suffix} \rangle, \langle \text{numeric } 3 \rangle)$
 - $\langle \text{numeric } 1 \rangle$ inference index
 - $\langle \text{numerics list} \rangle$ list of premise indices
 - $\langle \text{string } 1 \rangle$ left inference label *math-mode* \LaTeX code
 - $\langle \text{string } 2 \rangle$ right inference label *math-mode* \LaTeX code
 - $\langle \text{string } 3 \rangle$ left delimiter label *math-mode* \LaTeX code
 - $\langle \text{string } 4 \rangle$ right delimiter label *math-mode* \LaTeX code
 - $\langle \text{numeric } 2 \rangle$ junction style (0, 1, 2 or 3)
 - 0 fully-interlacing
 - 1 semi-interlacing
 - 2 non-interlacing
 - 3 user specified (tricky)
 - $\langle \text{suffix} \rangle$ alignment style (l, c, r or u)
 - l left
 - c centered
 - r right
 - u user specified (tricky)
 - $\langle \text{numeric } 3 \rangle$ inference line style (0, 1, 2, 3, 4, 5 or 6)


```

beginfig(370)
jgm 0 "a";
NFR 0 () ("1", "2", "3", "4", 1, c, 1);
draw drv_tree;
endfig;

```

$$3 \left\{ \frac{1}{a} \right\}^2 4$$

- “DCL $\langle \text{numeric } 1 \rangle$ ($\langle \text{numerics list} \rangle$)
 $\langle \text{string } 1 \rangle$, $\langle \text{string } 2 \rangle$, $\langle \text{string } 3 \rangle$, $\langle \text{string } 4 \rangle$)
 $\langle \text{numeric } 2 \rangle$, $\langle \text{suffix} \rangle$, $\langle \text{numeric } 3 \rangle$) $\langle \text{strings list} \rangle$;

is a shorthand for:

```

“jgm  $\langle \text{numeric } 1 \rangle$   $\langle \text{strings list} \rangle$ ;
NFR  $\langle \text{numeric } 1 \rangle$  ( $\langle \text{numerics list} \rangle$ )
 $\langle \text{string } 1 \rangle$ ,  $\langle \text{string } 2 \rangle$ ,  $\langle \text{string } 3 \rangle$ ,  $\langle \text{string } 4 \rangle$ )
 $\langle \text{numeric } 2 \rangle$ ,  $\langle \text{suffix} \rangle$ ,  $\langle \text{numeric } 3 \rangle$ );”

```

```

beginfig(380)
DCL 0 () ("1", "", "", "", 1, c, 1) "a";
DCL 1 () ("", "2", "", "", 1, c, 1) "b";
DCL 2 (0, 1) ("", "", "3", "", 1, c, 1) "c";
DCL 3 (2) ("", "", "", "4", 1, c, 1) "d";
draw drv_tree;
endfig;

```

$$3 \left\{ \frac{1}{\frac{a}{b}} \right\}^2 \frac{c}{d} 4$$

- MVD $\langle \text{numeric } 1 \rangle$ ($\langle \text{numeric } 2 \rangle$, $\langle \text{string } 1 \rangle$, $\langle \text{string } 2 \rangle$)
 $\langle \text{suffix} \rangle$, $\langle \text{numeric } 3 \rangle$)
- $\langle \text{numeric } 1 \rangle$ index of the origin judgment
 $\langle \text{numeric } 2 \rangle$ number of inference steps
 $\langle \text{string } 1 \rangle$ left phantom steps label *math-mode* L^AT_EX code
 $\langle \text{string } 2 \rangle$ right phantom steps label *math-mode* L^AT_EX code
 $\langle \text{suffix} \rangle$ alignment style (l, c, r or u)
 $\langle \text{numeric } 3 \rangle$ phantom steps style (0, 1, 2, 3, 4, 5 or 6)

```

beginfig(390)
jgm 0 "aaaaa";
jgm 1 "aaa";
jgm 2 "a";
jgm 3 "aaaaaaaaa";
jgm 4 "a";
nfr 0 () ("", 1);
nfr 1 () ("", 1);
nfr 2 (MVD 0 (2, "", "", 1, 3), 1) ("", 1);
nfr 3 () ("", 1);
nfr 4 (2, MVD 3 (5, "", "", r, 4)) ("", 1);
draw drv_tree;
endfig;

```

$$\begin{array}{c}
 \overline{\text{aaaaaaaaa}} \\
 \overline{\text{aaaaa}} \\
 \text{aaa} \\
 \overline{\text{a}} \\
 \text{a}
 \end{array}$$


```

beginfig(391)
jgm 0 "\textsc{Size matters -- Part 2}";
jgm 1 "\text{Here is an even longer judgment"&
      " that I don't want to shorten either.}";
jgm 2 "\text{This time I'm pretty sure that the"&
      " derivation tree won't fit on the page.}";
jgm 3 "\text{It does! Amazing.}";
nfr 0 () ("", 0);
nfr 1 (0) ("", 1);
nfr 2 () ("", 1);
nfr 3 (MVD 1 (2, "", "", 1, 3), 2) ("", 1);
draw drv_tree;
endfig;

```

S	– P	2
Here is an even longer judgment that I don't want to shorten either.		
This time I'm pretty sure that the derivation tree won't fit on the page.		
It does! Amazing.		

4.2 Inside derivation trees

Components and central points

```

beginfig(400) % components
jgm 7 "A", "B";
NFR 7 () ("(1)", "(2)", "(3)", "(4)", 0, c, 1);
drv_freeze; % usually called by drv_tree
draw sbj[7][0] withcolor (0, 0, 1); % sub-judgment      A
draw sbj[7][1] withcolor (0, 1, 0); % sub-judgment      B
draw l_ilb[7]  withcolor (0, 1, 1); % left inference label (1)
draw r_ilb[7]  withcolor (1, 0, 0); % right inference label (2)
draw l_dlb[7]  withcolor (1, 0, 1); % left delimiter label (3)
draw r_dlb[7]  withcolor (1, 1, 0); % right delimiter label (4)
draw l_dlm[7]; % left delimiter
draw iln[7];   % inference line
draw r_dlm[7]; % right delimiter
endfig;

```

$$(3) \left\{ \overset{(1)}{\overline{AB}} \overset{(2)}{\quad} \right\} (4)$$


```

beginfig(401) % central points
jgm 7 "A", "B";
NFR 7 () ("(1)", "(2)", "(3)", "(4)", 0, c, 1);
draw drv_tree withcolor .75*background;
pickup pencircle scaled 2.5;
draw sbj[7][0].c withcolor (0, 0, 1);
draw sbj[7][1].c withcolor (0, 1, 0);
draw l_ilb[7].c withcolor (0, 1, 1);
draw r_ilb[7].c withcolor (1, 0, 0);
draw l_dlb[7].c withcolor (1, 0, 1);
draw r_dlb[7].c withcolor (1, 1, 0);
endfig;

```

$$(3) \left\{ \begin{matrix} (1) \\ \overline{AB} \\ (2) \end{matrix} \right\} (4)$$


Dimensions

```

beginfig(410)
dcl 0 () ("", 0) "\gamma";
dcl 1 (0) ("", 1) "\delta";
draw drv_tree;
endfig;

```

(The picture below may look weird if you don't use scalable fonts.)



Depths (`den_dp`, `jdg[0].dp` and `jdg[1].dp`) are negative; heights (`num_hg`, `jdg[0].hg` and `jdg[1].hg`) are positive.

`drv.mp` typesets derivation trees essentially according to the algorithm for typesetting fractions described in Appendix G of the \TeX book (see [2, 3]).

drv_axis `drv_axis (<suffix>, <numeric>)`
 <suffix> reference type (`iln`, `jdg` or `dln`)
 `iln` inference line
 `jdg` judgment
 `dln` delimiter
 <numeric> reference judgment index

```
beginfig(420)
dcl 1 () ("", 1) "a";
Dcl 2 (1) ("", "{", "{", 1) "b";
draw drv_tree;
drv_axis (iln, 1);
resp. drv_axis (jdg, 2);
resp. drv_axis (dln, 2);
endfig;
```

$$\text{---math axis---} \left\{ \frac{\overline{a}}{b} \right\} - \left\{ \frac{\overline{a}}{b} \right\} - \left\{ \frac{\overline{a}}{b} \right\} \text{---math axis---}$$

4.3 Basic METAPOST use

“User specified” junction style

```
beginfig(430)
jgm 0 "{\cdot}";
jgm 1 "{\cdot}";
jgm 2 "\text{You may check that the distance"&
      " between the two dots above is 5cm.}";
nfr 0 () ("", 1);
nfr 1 () ("", 1);
NFR 2 (0, 1) ("", "", "", "", 3, c, 1); % caution: 3
xpart jdg[1].c=xpart jdg[0].c+5cm;
draw drv_tree;
endfig;
```

$$\overline{\cdot} \qquad \qquad \qquad \overline{\cdot}$$

You may check that the distance between the two dots above is 5cm.

“User specified” alignment style

```

beginfig(440)                                     % "\vdash":
jgm 0 "B, A, \Gamma", "\vdash", "C";              % sbj[0][1]
jgm 1 "A, \Gamma", "\vdash", "B\multimap C";      % sbj[1][1]
jgm 2 "\Gamma", "\vdash", "A\multimap(B\multimap C)"; % sbj[2][1]
nfr 0 () ("", 0);
NFR 1 (0) ("", "\multimap_{R}", "", "", 0, u, 1); % caution: u
NFR 2 (1) ("", "\multimap_{R}", "", "", 0, u, 1); % caution: u
xpart sbj[0][1].c=xpart sbj[1][1].c=xpart sbj[2][1].c;
draw drv_tree;
endfig;

```

$$\frac{\frac{B, A, \Gamma \vdash C}{A, \Gamma \vdash B \multimap C} \multimap_R}{\Gamma \vdash A \multimap (B \multimap C)} \multimap_R$$

drv_styled

$\langle path \rangle$ drv_styled $\langle numeric \rangle$
 $\langle path \rangle$ METAPOST path expression
 $\langle numeric \rangle$ path style (0, 1, 2, 3, 4, 5 or 6)

```

beginfig(450)
jgm 4 "A", "\vdash", "A";
jgm 5 "B", "\vdash", "B";
jgm 6 "A", " , ", "A", "\multimap", "B", "\vdash", "B";
jgm 7 "A", "\multimap", "B", "\vdash", "A", "\multimap", "B";
nfr 4 () ("1", 1);
nfr 5 () ("1", 1);
nfr 6 (4, 5) ("\multimap_{L}", 1);
nfr 7 (6) ("\multimap_{R}", 1);
draw drv_tree;
draw (sbj[7][2].c shifted (0, -num_hg) ..
      sbj[7][2].c {up} ..
      sbj[6][4].c ..
      sbj[5][0].c .. tension 1.05 ..
      sbj[5][2].c ..
      sbj[6][6].c ..
      sbj[7][6].c {down} ..
      sbj[7][6].c shifted (0, -num_hg))
      drv_styled 2 withcolor (159, 182, 205)/255; % slategray3
endfig;

```


$$\frac{\frac{\overline{A \vdash A}^1 \quad \overline{B \vdash B}^1}{A, A \multimap B \vdash B}^{\multimap_L} \quad \frac{}{A \multimap B \vdash A \multimap B}^{\multimap_R}$$

References

- [1] John D. Hobby. *A User's Manual for METAPOST*, 2009.
<http://tug.org/metapost.html>
- [2] Bogusław Jackowski. *Appendix G illuminated*. *TUGboat*, 27(1):83–90, 2006.
<http://www.tug.org/TUGboat/Articles/tb27-1/>
- [3] Donald E. Knuth. *The T_EXbook*. Addison-Wesley, 1984.

A Debugging

Recall that you have to run “`mpost <jobname>.mp`” *at least twice* (once more if you use sub-tree delimiters). If you get an error on the first run then it comes from the `drv.mp`/METAPOST code. If you get an error on the second run then it comes from the \LaTeX code.

Error on the first run METAPOST behaves essentially as \TeX / \LaTeX when it finds an error (see [1, Debugging]): it stops, “explains” the error in some way (look for the line starting with an exclamation mark), shows some lines of context, and asks you what to do next (answer `x` to terminate the run). If you’re lucky, the error comes from an inconsistency that `drv.mp` can detect: in such a case the explanation should be quite understandable (see below).

<pre> 50 beginfig(460) 51 jgm 0 "A\vdash B"; 52 jgm 1 "B\vdash C"; 53 jgm 2 "A\vdash C"; 54 jgm 3 "C\vdash D"; 55 jgm 4 "A\vdash D"; 56 nfr 0 () ("f", 1); 57 nfr 1 () ("g", 1); 58 nfr 2 (0, 1) ("\circ", 1); 59 nfr 3 () ("h", 1); 60 nfr 4 (0, 3) ("\circ", 1); 61 draw drv_tree; 62 endfig; </pre>	<p>METAPOST error message.</p> <pre> ! drv (fig. 460): 0 has been used already as a premise index (inference 2). <error context> 1.60 nfr 4 (0, 3) ("\circ", 1) ; ? </pre>
--	--

Error on the second run METAPOST fails to preprocess the \LaTeX code in $\langle jobname \rangle$ -delayed.mp and suggests that you “see `mpxerr.log`”, which is a regular \LaTeX log-file. This file tells you which part of the \LaTeX code is faulty but unfortunately not where to find it in $\langle jobname \rangle$.mp.

B Related packages

- [bussproofs.sty](#) (Samuel R. B.);
- [proof.sty](#) (Makoto T.);
- [prooftree.sty](#) (Paul T.);
- [virginialake.sty](#) (Alessio G.).

Some of these packages are described on Peter S.’s [\$\text{\LaTeX}\$ for Logicians](#) web-page.